


Practice #3

アニメーション (時間的な処理)

演習3A 時間処理・イベント処理・変数のスコープ



Processingをはじめよう
第二版
(Make: PROJECTS)

オンデマンド授業 6.06

課題学習 6.13

p. 64 - 71

p. 236 - 239

アニメーションの基礎：setup()とdraw()の関係

```
int count; //カウンター

//初回の処理
void setup(){
  size(500,500); //サイズ
  frameRate(10); //フレームレート

  textSize(200); //テキストサイズ
  fill(0); stroke(0); //線・文字は黒


  count = 0; //カウンター初期化
}

//定期的な処理
void draw(){
  background(255); //白く塗りつぶす

  //カウンターを書く
  text(count, 150, 300);
  // 罫線
  line(150,0,150,height);
  line(0,300,width,300);

  count++; //カウンターを更新
}
```

sample3A_1.pde



実行結果

void text(text, x, y);

text (文字列、または数字) を, (x, y) に表示.

void textSize(n);

text関数実行時のサイズを n に指定

関数間で共有すべき変数の宣言

```
int count; //カウンター
```

グローバル変数の宣言

初期化処理（初回に一回だけ実行）

```
void setup(){  
  size(500,500); //サイズ  
  frameRate(10); //フレームレ  
  textSize(200); //テキストサ  
  fill(0); stroke(0); //線・文  
  count = 0; //  
}
```

setup関数

setup関数を実行後、定期的に行う処理

```
void draw(){  
  background(255); //白く塗りつぶす  
  
  //カウンターを書く  
  text(count,150,300);  
  //罫線  
  line(150,0,150,height);  
  line(0,300,width,300);  
  
  count++; //カウンター  
}
```

draw関数


```
void frameRate(fps);
```

1秒間に何回draw関数を実行するかを決めます。デフォルトは30fps

1000/fps
[ms]

1000/fps
[ms]

1000/fps
[ms]



(count=0)

background(255);



//カウンターを書く
text(count, 150, 300);

0

line(150, 0, 150, height);
line(0, 300, width, 300);

count++;

(count=1)



(count=1)

background(255);



//カウンターを書く
text(count, 150, 300);

1

line(150, 0, 150, height);
line(0, 300, width, 300);

count++;

(count=2)



//定期的な処理

void draw(){

background(255); //白く塗りつぶす

//カウンターを書く

text(count, 150, 300);

//罫線

line(150, 0, 150, height);

line(0, 300, width, 300);

count++; //カウンターを更新

}

(count=2)

background(255);



//カウンターを書く
text(count, 150, 300);

2

line(150, 0, 150, height);
line(0, 300, width, 300);

Processingにあらかじめ用意されている関数

変更できない関数の例

戻り値のない（計算結果のない）関数

```
void rect(x,y,w,h); void line(x1,y1,x2,y2);
```

```
void fill(red,green,blue); void stroke(red,green,blue);
```

戻り値のある（計算結果のある）関数

```
float dist(x1,y1,x2,y2); float pow(x,n); float sqrt(x);
```

(x1,y1) と (x2,y2) の距離

xのn乗

xの平方根

変更することが前提とされている関数の例

以下の関数は、初期状態では空白ですが、独自に定義することによって、アニメーション・イベント処理を埋め込むことができます。
また、「void」は、戻り値が無いことを意味します。

```
void setup();
```

プログラム起動時の処理

```
void draw();
```

setup関数終了後、定期的に行う処理

```
void mousePressed();
```

マウスが押された時の処理

```
void mouseReleased();
```

マウスが離された時の処理

変更することが前提とされている関数の例

```
void setup();
```

プログラム起動時の処理

```
void draw();
```

setup関数終了後, 定期的に行う処理

```
void mousePressed();
```

マウスが押された時の処理

```
void mouseReleased();
```


マウスが離された時の処理

これらの関数は, 初期状態では空白ですが, 独自に定義することによって, アニメーション・イベント処理を埋め込むことができます.
また, 「void」は, 返り値が無いことを意味します.

関数を定義するための記法


p.256

```
void setup(){
```




```
}
```

```
void draw(){
```




```
}
```

```
void mousePressed(){
```



```
}
```



```
void mouseReleased(){
```




```
}
```

{と}の間に, 必要な処理を書きます. 何も書かないということは, 何の処理もしないことを積極的に意味するので, エラーにはなりません.


マウスイベント




マウスをクリック



実行結果




マウスをリリース



sample3A_2.pde

「sample4A_1.pde」に以下
を書き加えてみてください。

```
//マウスを押された時の処理  
void mousePressed(){  
  fill(255,100,0); //色を変える  
}  
  
//マウスを離した時の処理  
void mouseReleased(){  
  fill(0); //色を黒に戻す  
  count = 0; //カウンターを戻す  
}
```



グローバル変数とローカル変数

sample3A_3.pde

```
int count;
```

```
void setup(){  
  count = 0;  
  frameRate(10);  
}
```

```
void draw(){  
  count++;  
}
```

```
void mousePressed(){  
  println(count);  
}
```

```
void setup(){  
  count = 0;  
  frameRate(10);  
}
```

```
void draw(){  
  count++;  
}
```

int count; **グローバル変数**

```
void mousePressed(){  
  println(count);  
}
```

6
12
20

COUNTは0から始まり、1秒間に10の割合で増えていきます。マウスを押すと、現在の数字が表示されます。

コンソール

関数の定義の外側で宣言された変数は「グローバル変数」と呼ばれ、その値は、全ての関数内で共有されます。宣言と同時に値を代入することもできます。

グローバル変数とローカル変数

sample3A_4.pde

```
void setup(){  
  int count = 0;  
  println(count);  
}
```

```
void draw(){  
  count++;  
}
```

文法エラー

```
void mousePressed(){  
  println(count);  
}
```

文法エラー

```
void setup(){
```

ローカル変数

```
int count = 0;
```

```
println(count);
```

```
}
```

```
void draw(){  
  count++;  
}
```



```
void mousePressed(){  
  println(count);  
}
```


特定の関数の内部で宣言された変数は、その関数の内部でのみ使用できます。(関数をまたいで)別の関数内で共有することはできません。

アニメーション

```
1 int count; //カウンター
2
3 void setup(){
4   size(500,500); //サイズ
5   //背景白、線なし
6   background(255); noStroke();
7   //50fps
8   frameRate(50); //50fps
9
10  //カウンターを0に初期化
11  count = 0;
12 }
```

```
14 void draw(){
15
16   for(int i=0;i<50;i++){
17     for(int j=0;j<50;j++){
18
19       float x = 1 + 10*i;
20       float y = 1 + 10*j;
21
22       if(j<count%50){
23         fill(255,100,0);
24       }else{
25         fill(0);
26       }
27       rect(x,y,8,8);
28     }
29   }
30   count++;
31 }
```


sample3A_5.pde




実行結果

```
14 void draw(){
15
16   for(int i=0;i<50;i++){
17     for(int j=0;j<50;j++){
18
19       float x = 1 + 10*i;
20       float y = 1 + 10*j;
21
22       if(j<count%50){
23         fill(255,100,0);
24       }else{
25         fill(0);
26       }
27       rect(x,y,8,8);
28     }
29   }
30   count++;
31 }
```

sample3A_5.pde



count	count % 50
0	0
↓	↓
49	49
50	0
↓	↓
99	49
100	0
↓	↓
	49
	0




練習

```
1 int count; //カウンター
2 float diagonal; //対角線の長さ
3
4 void setup(){
5     size(500,500); //サイズ
6     //背景白、線なし
7     background(255); noStroke();
8     //50fps
9     frameRate(50);
10
11     //対角線の距離を計算
12     diagonal = dist(0,0,width,height);
13     //カウンターを0に初期化
14     count = 0;
15 }
```

```
17 void draw(){
18     //ペースメーカー
19     float disP = count % diagonal;
20
21     for(int i=0;i<50;i++){
22         for(int j=0;j<50;j++){
23
24             float x = 1 + 10*i;
25             float y = 1 + 10*j;
26
27             if(dist(0,0,x,y)<disP){
28                 fill(255,100,0);
29             }else{
30                 fill(0);
31             }
32             rect(x,y,8,8);
33         }
34     }
35     count++;
36 }
```

sample3A_6.pde



dist(0,0,x,y)
(<disP)

dist(0,0,x,y)
(>disP)

ペースメーカー

float disP = count % diagonal;

対角線の長さ

diagonal = dist(0,0,width,height);

float dist(x1,y1,x2,y2);

(x1,y1)と(x2,y2)の距離を計算します。
sqrt((x2-x1)*(x2-x1)+(y2-y1)*(y2-y1))と同じです。

マウス座標の取得

mouseX

mouseY

マウスのX座標とY座標
(ウィンドウ領域内にある時のみ追尾)

```
1 int count; //カウンター
2 float cx = 0; float cy = 0; //円の中心座標
3 boolean mode = false; //描画の有無
4
5 void setup(){
6   size(500,500);
7   background(255); noStroke();
8   frameRate(50);
9   count = 0;
10 }
```

描画の有無の切り替え

(cx,cy)からの距離
がcountのときの
み色を橙色にそれ
以外は黒とします。

```
33 void mousePressed(){
34   cx = mouseX; cy = mouseY;
35   count = 0;
36   mode = true;
37 }
```

マウスが押されたた
場所を(cx,cy)とし、
countを初期化し、描
画モードとします。

```
39 void mouseReleased(){
40   mode = false;
41 }
```

マウスが離された時に、描
画モードをオフとします。

```
12 void draw(){
13
14   for(int i=0;i<50;i++){
15     for(int j=0;j<50;j++){
16
17       float x = 1 + 10*i;
18       float y = 1 + 10*j;
19
20       if(dist(cx,cy,x,y)<count){
21         fill(255,100,0);
22       }else{
23         fill(0);
24       }
25       rect(x,y,8,8);
26     }
27   }
28   if(mode){
29     count++;
30   }
31 }
```

modeがtrueのと
きのみ、countを増
やします。

sample3A_7.pde


sample3A_7.pde

マウスをクリック

マウスをクリック

マウスをリリース

実行結果



練習

「sample3A_7.pde」の一部を修正して、マウスを離してもこれまでの描画が消えずに蓄積していくように変更してみてください。

